

Faster Coroutine Pipelines: A Reconstruction (Extended Version)

Ruben P. Pieters
Tom Schrijvers

Report CW715, November 2018



KU Leuven
Department of Computer Science
Celestijnenlaan 200A – B-3001 Heverlee (Belgium)

Faster Coroutine Pipelines: A Reconstruction (Extended Version)

Ruben P. Pieters

Tom Schrijvers

Report CW 715, November 2018

Department of Computer Science, KU Leuven

Abstract

Spivey has recently presented a novel functional representation that supports the efficient composition, or merging, of coroutine pipelines for processing streams of data. This representation was inspired by Shivers and Might’s three-continuation approach and is shown to be equivalent to a simple yet inefficient executable specification. Unfortunately, neither Shivers and Might’s original work nor the equivalence proof sheds much light on the underlying principles allowing the derivation of this efficient representation from its specification.

This paper gives the missing insight by reconstructing a systematic derivation in terms of known transformation steps from the simple specification to the efficient representation. This derivation sheds light on the limitations of the representation and on its applicability to other settings. In particular, it has enabled us to obtain a similar representation for pipes featuring two-way communication, similar to the Haskell pipes library. Our benchmarks confirm that this two-way representation retains the same improved performance characteristics.

Faster Coroutine Pipelines: A Reconstruction (Extended Version)

Ruben P. Pieters^[0000–0003–0537–9403] and Tom Schrijvers^[0000–0001–8771–5559]

KU Leuven, 3001 Leuven, Belgium
{ruben.pieters, tom.schrijvers}@cs.kuleuven.be

Abstract. Spivey has recently presented a novel functional representation that supports the efficient composition, or *merging*, of coroutine pipelines for processing streams of data. This representation was inspired by Shivers and Might’s three-continuation approach and is shown to be equivalent to a simple yet inefficient executable specification. Unfortunately, neither Shivers and Might’s original work nor the equivalence proof sheds much light on the underlying principles allowing the derivation of this efficient representation from its specification.

This paper gives the missing insight by reconstructing a systematic derivation in terms of known transformation steps from the simple specification to the efficient representation. This derivation sheds light on the limitations of the representation and on its applicability to other settings. In particular, it has enabled us to obtain a similar representation for pipes featuring two-way communication, similar to the Haskell `pipes` library. Our benchmarks confirm that this two-way representation retains the same improved performance characteristics.

Keywords: Stream Processing · Structured Recursion · Algebra.

1 Introduction

Coroutine pipelines provide a compositional approach to processing streams of data that is both efficient in time and space, thanks to a targeted form of lazy evaluation interacting well with side-effects like I/O. Two prominent Haskell libraries for coroutine pipelines are `pipes` [5] and `conduit` [12]. Common to both libraries is their representation of pipelines by an algebraic data type (ADT).

Spivey [13] has recently presented a novel Haskell representation that is entirely function-based. His representation is an adaptation of Shivers and Might’s earlier three-continuation representation [11] and exhibits a very efficient *merge* operation for connecting pipes.

Spivey proves that his representation is equivalent to a simple ADT-based specification. Yet, neither his proof nor Shivers and Might’s explanation sheds much light on the underlying principles used to come up with the efficient representation. This makes it difficult to adapt the representation to other settings.

This paper remedies the situation by systematically deriving the efficient function-based representation from the simple, but inefficient ADT-based representation. Our derivation consists of known transformations and constructions

that are centered around folds with appropriate algebras. Our derivation clarifies the limitations of the efficient representation, and enables us to derive a similarly efficient representation for the two-way pipes featured in the `pipes` library.

The specific contributions of this paper are:

- We present a systematic derivation of Spivey’s efficient representation starting from a simple executable specification. Our derivation only consists of known transformations, most of which concern structural recursion with folds and algebras. It also explains why the efficient representation only supports the merging of “never-returning” pipes.
- We apply our derivation to a more general definition of pipes used by the `pipes` library, where the communication between adjacent pipes is bidirectional rather than unidirectional.
- Our benchmarks demonstrate that the merge operator for the bidirectional three-continuation approach improves upon the `pipes` library’s performance.

The rest of this paper is organized as follows. Section 2 briefly introduces both the ADT pipes encoding and the three-continuation approach. Section 3 derives the fast merging operation for a simplified setting. Section 4 derives the fast merging operation for the original pipe setting. Section 5 extends Spivey’s approach with the bidirectional pipes operations. Section 6 presents the results of the primes benchmark by Spivey, on the approaches discussed in this paper. Section 7 discusses related work and Section 8 concludes this paper.

2 Motivation

This section introduces the ADT pipes encoding and then contrasts it with the three-continuation encoding. This serves as both a background introduction and a motivation for a better understanding of the relation between both encodings.

2.1 Pipes

We start with a unidirectional version of the `pipes` library. A unidirectional *pipe* can receive i values, output o values and return a values. On the other hand, a bidirectional pipe additionally carries an output value when receiving values and an input value when outputting values. We represent a unidirectional pipe as an abstract syntax tree where each node is an input, output or return operation. This is expressed in Haskell with the following ADT.

$$\begin{aligned} \text{data } \text{Pipe } i \ o \ a = & \text{Input } (i \rightarrow \text{Pipe } i \ o \ a) \\ & | \text{Output } o \ (\text{Pipe } i \ o \ a) \\ & | \text{Return } a \end{aligned}$$

This datatype exhibits a monadic structure where the bind operation (\gg) grafts one syntax tree onto another.

instance *Monad* (*Pipe* *i* *o*) **where**
return = *Return*
(*Input* *h*) $\gg=$ *f* = *Input* ($\lambda i \rightarrow (h\ i) \gg= f$)
(*Output* *o* *r*) $\gg=$ *f* = *Output* *o* (*r* $\gg=$ *f*)
(*Return* *a*) $\gg=$ *f* = *f* *a*

We define the basic components: *input_P*: a pipe returning its received input, *output_P*: a pipe outputting a set value and *return_P*: a pipe returning a set value.

input_P :: *Pipe* *i* *o* *i*
input_P = *Input* ($\lambda i \rightarrow$ *Return* *i*)
output_P :: *o* \rightarrow *Pipe* *i* *o* ()
output_P *o* = *Output* *o* (*Return* ())
return_P :: *a* \rightarrow *Pipe* *i* *o* *a*
return_P *a* = *Return* *a*

The bind operation assembles these components into larger pipes. For example *doubler_P*, a pipe which repeatedly takes its input, multiplies it by two and continually outputs this new value.

doubler_P :: *Pipe* *Int* *Int* *a*
doubler_P = **do** *i* \leftarrow *input_P*; *output_P* (*i* * 2); *doubler_P*

Another essential way of combining pipes is *merging* them. This connects the outputs of the upstream to the inputs of the downstream. In the implementation, *merge_{PL}* performs a case analysis on the downstream *q*: if it is trying to output then that is kept and we keep searching, if it finds an input then we call *merge_{PR}* on the wrapped continuation and the upstream. Then, in *merge_{PR}* we similarly scan the upstream for an output operation, keeping any input operations. If an output operation is found, the output value is passed to the continuation and the merging process starts again. If at any point we see a return, then the merge finishes with this resulting return value. The implementation is given below.

merge_P :: *Pipe* *i* *m* *a* \rightarrow *Pipe* *m* *o* *a* \rightarrow *Pipe* *i* *o* *a*
merge_P *p* *q* = *merge_{PL}* *q* *p* **where**
merge_{PL} :: *Pipe* *m* *o* *a* \rightarrow *Pipe* *i* *m* *a* \rightarrow *Pipe* *i* *o* *a*
merge_{PL} (*Input* *h*) *p* = *merge_{PR}* *p* *h*
merge_{PL} (*Output* *o* *r*) *p* = *Output* *o* (*merge_{PL}* *r* *p*)
merge_{PL} (*Return* *a*) *p* = *Return* *a*
merge_{PR} :: *Pipe* *i* *m* *a* \rightarrow (*m* \rightarrow *Pipe* *m* *o* *a*) \rightarrow *Pipe* *i* *o* *a*
merge_{PR} (*Input* *f*) *h* = *Input* ($\lambda v \rightarrow$ *merge_{PR}* (*f* *v*) *h*)
merge_{PR} (*Output* *o* *r*) *h* = *merge_{PL}* (*h* *o*) *r*
merge_{PR} (*Return* *a*) *h* = *Return* *a*

The merge operator enables expressing the merge of *doubler_P* with itself. In this example the left *doubler_P* is the upstream and the right *doubler_P* is the

downstream. The result of this merge is a pipe which outputs the quadruple of its incoming values.

```
quadruplerP :: Pipe Int Int a
quadruplerP = doublerP 'mergeP' doublerP
```

We can run a pipe by interpreting it to *IO*.

```
toIOP :: (Read i, Show o) => Pipe i o a -> IO a
toIOP (Input f)      = do i <- readLn; toIOP (f i)
toIOP (Output o r) = do putStrLn ("out: " ++ show o); toIOP r
toIOP (Return a) = return a
```

An example where we input *10*, receive *40* and then exit, is shown below.

```
λ > toIOP quadruplerP
10 <Return>
out : 40
<Ctrl+C>
```

2.2 Three-Continuation Approach

The function *merge_P* is suboptimal because it has to recursively scan a pipe for an operation of interest while copying the other operation. When several merges are piled up this leads to repeated scanning and copying of the same operations.

Spivey has introduced *ContPipe*, a different pipe representation which enables a faster merge implementation [13]. It features three continuations, one for each constructor. The first continuation ($a \rightarrow \text{Result } i \ o$) represents the return constructor. The next two continuations, *InCont* *i* and *OutCont* *o* as part of *Result* *i* *o*, represent the input and output constructor respectively.

```
newtype ContPipe i o a =
  MakePipe { runPipe :: (a -> Result i o) -> Result i o }
type Result i o = InCont i -> OutCont o -> IO ()
newtype InCont i =
  MakeInCont { resumeI :: OutCont i -> IO () }
newtype OutCont o =
  MakeOutCont { resumeO :: o -> InCont o -> IO () }
instance Monad (ContPipe i o) where
  return a = MakePipe (λk -> k a)
  p >>= f = MakePipe (λk -> runPipe p (λx -> runPipe (f x) k))
```

In the following definitions for the basic pipe components the continuation *k* is the return constructor—we give it a value and the input and output constructors and receive a pipe. The continuations *k_i* and *k_o* are the input and output constructors, we resume them with the newtype unwrapper and the continuations are refreshed once they have been used.

```

returnCP :: a → ContPipe i o a
returnCP a = MakePipe (λk ki ko → k a ki ko)

inputCP :: ContPipe i o i
inputCP = MakePipe (λk ki ko →
  resumeI ki (MakeOutCont (λi k'i → k i k'i ko)))

outputCP :: o → ContPipe i o ()
outputCP o = MakePipe (λk ki ko →
  resumeO ko o (MakeInCont (λk'o → k () ki k'o)))

```

We can use the *Monad* instance for *ContPipe* to compose pipes with **do**-notation, similar to *Pipe*.

```

doublerCP :: ContPipe Int Int a
doublerCP = do i ← inputCP; outputCP (i * 2); doublerCP

```

We can also interpret *ContPipe* to *IO*.

```

toIOCP :: (Read i, Show o) ⇒ ContPipe i o () → IO ()
toIOCP p = runPipe p (λ() _ → return ()) ki ko where
  ki = MakeInCont (λko → do x ← readLn; resumeO ko x ki)
  ko = MakeOutCont (λo ki →
    do putStrLn ("out: " ++ show o); resumeI ki ko)

```

The merge function for *ContPipe* is defined as:

```

mergeCP p q = MakePipe (λk ki ko →
  runPipe q err (MakeInCont (λk'o → runPipe p err ki k'o)) ko)
where err = error "terminated"

```

With the merge definition we are able to create the quadrupler pipe as before. Running *toIO_{CP} quadrupler_{CP}* results an identical scenario to the *Pipe* scenario from the previous section.

```

quadruplerCP :: ContPipe Int Int a
quadruplerCP = doublerCP 'mergeCP 'doublerCP

```

While Spivey has demonstrated the remarkable performance advantage of this merge operator, he sheds little light on the origin or underlying principles of the related encoding. The remainder of this paper provides this missing insight by deriving Spivey's efficient *ContPipe* representation from the ADT-style *Pipe* by means of well-known principles. The aim is to improve understanding of the applicability and limitations of the techniques used.

3 Fast Merging for One-Sided Pipes

To offer a firmer grip on the problem, this section considers a simplified setting where pipes are one-sided, either only producing or only consuming data. For example, the *doubler* component can not be defined in this setting. The simplified setting gives a more straightforward path to the fast merging approach, which we generalize back to regular 'mixed' pipes in Section 4.

3.1 One-Sided Pipes

In the simplified setting pipes are either pure *Producers* or pure *Consumers*. A *Producer* only outputs values, while a *Consumer* only receives them.

data *Producer* $o = \text{Producer } o (\text{Producer } o)$
data *Consumer* $i = \text{Consumer } (i \rightarrow \text{Consumer } i)$

If we specialize merge_P for a *Consumer* and a *Producer*, we get:

$\text{merge}_A :: \text{Producer } b \rightarrow \text{Consumer } b \rightarrow a$
 $\text{merge}_A p q = \text{merge}_{AL} q p$ **where**
 $\text{merge}_{AL} :: \text{Consumer } b \rightarrow \text{Producer } b \rightarrow a$
 $\text{merge}_{AL} (\text{Consumer } h) p = \text{merge}_{AR} p h$
 $\text{merge}_{AR} :: \text{Producer } b \rightarrow (b \rightarrow \text{Consumer } b) \rightarrow a$
 $\text{merge}_{AR} (\text{Producer } o r) h = \text{merge}_{AL} (h o) r$

3.2 Mutual Recursion Elimination

The two auxiliary functions merge_{AL} and merge_{AR} turn respectively a producer and a consumer into the result of type a by means of an additional parameter, which is respectively of type $(\text{Producer } b)$ and $(b \rightarrow \text{Consumer } b)$. To highlight these parameters, we introduce type synonyms for them.

type *ProdPar'* $b = \text{Producer } b$
type *ConsPar'* $b = b \rightarrow \text{Consumer } b$

Now we refactor merge_{AL} and merge_{AR} with respect to their additional parameter in a way that removes the term-level mutual recursion between them. Consider merge_{AL} which does not use its parameter p directly, but only its interpretation by function merge_{AR} . We refactor this code to a form where merge_{AR} has already been applied to p before it is passed to merge_{AL} . This adapted merge_{AL} would then have type $\text{Consumer } b \rightarrow (\text{ConsPar}' b \rightarrow a) \rightarrow a$. At the same time we apply a similar transformation to merge_{AR} , moving the application of merge_{AL} to h out of it. This yields infinite types for the two new parameters, which Haskell only accepts if we wrap them in newtypes.

newtype *ProdPar* $b a = \text{ProdPar } (\text{ConsPar } b a \rightarrow a)$
newtype *ConsPar* $b a = \text{ConsPar } (b \rightarrow \text{ProdPar } b a \rightarrow a)$

The merge is then defined by appropriately placing newtype (un-)wrappers.

$\text{merge}_{Par} :: \text{Producer } b \rightarrow \text{Consumer } b \rightarrow a$
 $\text{merge}_{Par} p q = \text{ml } q (\text{ProdPar } (mr p))$ **where**
 $\text{ml} :: \text{Consumer } b \rightarrow \text{ProdPar } b a \rightarrow a$
 $\text{ml } (\text{Consumer } h) (\text{ProdPar } p) = p (\text{ConsPar } (\lambda i \rightarrow (\text{ml } (h i))))$
 $\text{mr} :: \text{Producer } b \rightarrow \text{ConsPar } b a \rightarrow a$
 $\text{mr } (\text{Producer } o r) (\text{ConsPar } h) = h o (\text{ProdPar } (mr r))$

Note that we can recover Spivey's *InCont* i and *OutCont* o by instantiating the type parameter a to $\text{IO } ()$ in $\text{ProdPar } i a$ and $\text{ConsPar } o a$ respectively.

3.3 Structural Recursion with Fold

Due to the removal of the term-level mutual recursion in *ml* and *mr*, they are easily adapted to their structurally recursive form. By isolating the work done in each recursive step, we obtain *alg_L* and *alg_R*.

```

type CarrierL i a = ProdPar i a → a
algL :: (i → CarrierL i a) → CarrierL i a
algL f = λ(ProdPar prod) → prod (ConsPar f)
type CarrierR o a = ConsPar o a → a
algR :: o → CarrierR o a → CarrierR o a
algR o prod = λ(ConsPar cons) → cons o (ProdPar prod)

```

The functions *alg_L* and *alg_R* are now in a form known as algebras. Algebras are a combination of a *carrier* *r*, the type of the resulting value, and an *action* of type *f r* → *r*. This action denotes the computation performed at each node of the recursive datatype, for which the functor *f* determines the shape of its nodes. We omit the carrier type if it is clear from the context and simply refer to an algebra by its action.

The structural recursion schemes, or *folds*, for *Consumer* and *Producer* take algebras of the form (*i* → *r*) → *r* and *o* → *r* → *r*. Their definitions are:

```

foldP :: (o → r → r) → Producer o → r
foldP alg (Producer o r) = alg o (foldP alg r)
foldC :: ((i → r) → r) → Consumer i → r
foldC alg (Consumer h) = alg (λi → foldC alg (h i))

```

An example use of folds is an interpretation to *IO* by supplying the inputs for a consumer or printing the outputs of a producer.

```

type CarrierConsIO i = IO ()
consumeIO :: Read i ⇒ Consumer i → IO ()
consumeIO = foldC alg where
  alg :: Read i ⇒ (i → CarrierConsIO i) → CarrierConsIO i
  alg f = do x ← readLn; f x
type CarrierProdIO o = IO ()
produceIO :: Show o ⇒ Producer o → IO ()
produceIO = foldP alg where
  alg :: Show o ⇒ o → CarrierProdIO o → CarrierProdIO o
  alg o p = do print o; p

```

Another example is expressing *merge_{Par}* with folds using *alg_L* and *alg_R*.

```

mergefold :: Producer x → Consumer x → a
mergefold p q = foldC algL q (ProdPar (foldP algR p))

```

3.4 A Short-Cut to a Merge-Friendly Representation

Instead of directly defining a *Consumer* or *Producer* value in terms of the data constructors of the respective types, we can also do it in a more roundabout way by abstracting over the constructor occurrences—this is known as *build* form. The *build* function then instantiates the abstracted constructors with the actual constructors; for *Consumer* and *Producer* they are:

$$\begin{aligned} \text{build}_C &:: (\forall r. (i \rightarrow r) \rightarrow r) \rightarrow \text{Consumer } i \\ \text{build}_C g &= g \text{ Consumer} \\ \text{build}_P &:: (\forall r. (o \rightarrow r \rightarrow r) \rightarrow r) \rightarrow \text{Producer } o \\ \text{build}_P g &= g \text{ Producer} \end{aligned}$$

For instance,

$$\begin{aligned} \text{prodFrom} &:: \text{Integer} \rightarrow \text{Producer Integer} \\ \text{prodFrom } n &= \text{Producer } n (\text{prodFrom } (n + 1)) \end{aligned}$$

can be written as:

$$\begin{aligned} \text{prodFrom } n &= \text{build}_P (\text{prodFrom}' n) \text{ where} \\ \text{prodFrom}' &:: \text{Integer} \rightarrow (\forall r. (\text{Integer} \rightarrow r \rightarrow r) \rightarrow r) \\ \text{prodFrom}' n \ p &= \text{go } n \text{ where } \text{go } n = p \ n (\text{go } (n + 1)) \end{aligned}$$

The motivation for these build functions is use of the fold/build fusion rule, a special form of short-cut fusion [4]. This rule can be applied when a fold directly follows a build, specifically for *Consumer* and *Producer* these fusion rules are:

$$\begin{aligned} \text{fold}_C \text{ alg } (\text{build}_C \text{ cons}) &= \text{cons alg} \\ \text{fold}_P \text{ alg } (\text{build}_P \text{ prod}) &= \text{prod alg} \end{aligned}$$

In other words, instead of first building an ADT representation and then folding it to its result, we can directly create the result of the fold. This readily applies to the two folds in *merge_{fold}*. We can directly represent consumers and producers in terms of the carrier types of those two folds,

$$\begin{aligned} \text{type Consumer}_{Alt} \ i &= \forall a. \text{Carrier}_L \ i \ a \quad -- \forall a. \text{ProdPar } i \ a \rightarrow a \\ \text{type Producer}_{Alt} \ o &= \forall a. \text{Carrier}_R \ o \ a \quad -- \forall a. \text{ConsPar } o \ a \rightarrow a \end{aligned}$$

using their algebras as constructors:

$$\begin{aligned} \text{input}_{Alt} &:: (i \rightarrow \text{Consumer}_{Alt} \ i) \rightarrow \text{Consumer}_{Alt} \ i \\ \text{input}_{Alt} &= \text{alg}_L \\ \text{output}_{Alt} &:: o \rightarrow \text{Producer}_{Alt} \ o \rightarrow \text{Producer}_{Alt} \ o \\ \text{output}_{Alt} &= \text{alg}_R \end{aligned}$$

For instance, after fold/build fusion *prodFrom* becomes:

$$\begin{aligned} \text{prodFrom}_{Alt} &:: \text{Integer} \rightarrow \text{Producer}_{Alt} \text{ Integer} \\ \text{prodFrom}_{Alt} \, n &= \text{output}_{Alt} \, n \, (\text{prodFrom}_{Alt} \, (n + 1)) \end{aligned}$$

The merge function for the alternate representations Producer_{Alt} and Consumer_{Alt} then becomes an almost trivial operation.

$$\begin{aligned} \text{merge}_{Alt} &:: \text{Producer}_{Alt} \, b \rightarrow \text{Consumer}_{Alt} \, b \rightarrow a \\ \text{merge}_{Alt} \, p \, q &= q \, (\text{ProdPar} \, p) \end{aligned}$$

3.5 A Not So Special Representation

This merge-friendly representations of producers and consumers are not just specializations; they are in fact isomorphic to the originals. The inverses of ml and mr to complete the isomorphism are given by ml^{-1} and mr^{-1} . Appendix A proves the isomorphism.

$$\begin{aligned} ml^{-1} &:: \text{Consumer}_{Alt} \, i \rightarrow \text{Consumer} \, i \\ ml^{-1} \, f &= f \, (\text{ProdPar} \, h) \text{ where} \\ h &:: \text{ConsPar} \, i \, (\text{Consumer} \, i) \rightarrow \text{Consumer} \, i \\ h \, (\text{ConsPar} \, f) &= \text{Consumer} \, (\lambda x \rightarrow f \, x \, (\text{ProdPar} \, h)) \\ mr^{-1} &:: \text{Producer}_{Alt} \, o \rightarrow \text{Producer} \, o \\ mr^{-1} \, f &= f \, (\text{ConsPar} \, (\lambda x \, p \rightarrow \text{Producer} \, x \, (h \, p))) \text{ where} \\ h &:: \text{ProdPar} \, o \, (\text{Producer} \, o) \rightarrow \text{Producer} \, o \\ h \, (\text{ProdPar} \, f) &= f \, (\text{ConsPar} \, (\lambda x \, p \rightarrow \text{Producer} \, x \, (h \, p))) \end{aligned}$$

Hence, we can also fold with other algebras by transforming the merge-friendly representation back to the ADT, and then folding over that.

$$\begin{aligned} \text{fold}_{P_{Alt}} &:: (o \rightarrow a \rightarrow a) \rightarrow \text{Producer}_{Alt} \, o \rightarrow a \\ \text{fold}_{P_{Alt}} \, \text{alg} \, \text{rep} &= \text{fold}_P \, \text{alg} \, (mr^{-1} \, \text{rep}) \\ \text{fold}_{C_{Alt}} &:: ((i \rightarrow a) \rightarrow a) \rightarrow \text{Consumer}_{Alt} \, i \rightarrow a \\ \text{fold}_{C_{Alt}} \, \text{alg} \, \text{rep} &= \text{fold}_C \, \text{alg} \, (ml^{-1} \, \text{rep}) \end{aligned}$$

Of course, these definitions are wasteful because they create intermediate datatypes. However, by performing fold/build fusion we obtain their fused versions:

$$\begin{aligned} \text{fold}_{P_{Alt}} \, \text{alg} \, \text{rep} &= \text{rep} \, (\text{ConsPar} \, (\lambda x \, p \rightarrow \text{alg} \, x \, (h \, p))) \text{ where} \\ h \, (\text{ProdPar} \, f) &= f \, (\text{ConsPar} \, (\lambda x \, p \rightarrow \text{alg} \, x \, (h \, p))) \\ \text{fold}_{C_{Alt}} \, \text{alg} \, \text{rep} &= \text{rep} \, (\text{ProdPar} \, h) \text{ where} \\ h \, (\text{ConsPar} \, f) &= \text{alg} \, (\lambda x \rightarrow f \, x \, (\text{ProdPar} \, h)) \end{aligned}$$

4 Return to Two-Sided Pipes

The previous section has derived an efficient approach for simplified Consumer and Producer pipes. This section extends that approach to proper Pipes in two steps, first supporting both input and output operations, and then also a return .

4.1 Pipe of No Return

Let us consider pipes with both input and output operations, but no *return*.

data $Pipe_\infty$ i $o = Input_\infty (i \rightarrow Pipe_\infty i o)$
 $| Output_\infty o (Pipe_\infty i o)$

We can fold over these pipes by providing algebras for both the input and output operation, agreeing on the carrier type a .

$foldPipe_\infty :: Pipe_\infty i o \rightarrow ((i \rightarrow a) \rightarrow a) \rightarrow (o \rightarrow a \rightarrow a) \rightarrow a$
 $foldPipe_\infty p inAlg outAlg = go\ p\ \textbf{where}$
 $go (Input_\infty p) = inAlg (\lambda i \rightarrow go (p\ i))$
 $go (Output_\infty o p) = outAlg o (go\ p)$

To merge these pipes, we use alg_L and alg_R developed in the previous section. There is only one snag: the two algebras do not agree on the carrier type. The carrier types were the alternate representations $Consumer_{Alt}$ and $Producer_{Alt}$.

type $Consumer_{Alt}$ $i = \forall a. ProdPar\ i\ a \rightarrow a$
type $Producer_{Alt}$ $o = \forall a. ConsPar\ o\ a \rightarrow a$

We reconcile these two carrier types by observing that both are functions with a common result type, but different parameter types. A combination of both is a function taking both parameter types as input.

type $Result_R$ $i\ o = \forall a. ConsPar\ o\ a \rightarrow ProdPar\ i\ a \rightarrow a$

The algebra actions are easily adapted to the additional parameter. They simply pass it on to the recursive positions without using it themselves.

$input_{Result_R} :: (i \rightarrow Result_R i o) \rightarrow Result_R i o$
 $input_{Result_R} f = \lambda cons (ProdPar\ prod) \rightarrow$
 $prod (ConsPar (\lambda i\ prod' \rightarrow f\ i\ cons\ prod'))$
 $output_{Result_R} :: o \rightarrow Result_R i o \rightarrow Result_R i o$
 $output_{Result_R} o result = \lambda (ConsPar\ cons) prod \rightarrow$
 $cons\ o (ProdPar (\lambda cons' \rightarrow result\ cons'\ prod))$

Like before, we can avoid the algebraic datatype $Pipe_\infty$ and directly work with $Result_R$ using the algebras as constructor functions.

Finally, we can use the one-sided merge function from the previous section to merge the output side of a $Result_R\ i\ m$ pipe with the input side of a $Result_R\ m\ o$ pipe. Because we defer the interpretation of the i and o sides of the respective pipes, this one-sided merge does not yield a result of type a , but rather one of type $ConsPar\ o\ a \rightarrow ProdPar\ i\ a \rightarrow a$. In other words, the merge of the two pipes yields a $Result_R\ i\ o$ pipe.

$merge_{Result_R} :: Result_R i m \rightarrow Result_R m o \rightarrow Result_R i o$
 $merge_{Result_R} p\ q = \lambda cons_o\ prod_i \rightarrow$
 $\textbf{let } q' = q\ cons_o$
 $p' = flip\ p\ prod_i$
 $\textbf{in } q' (ProdPar\ p')$

4.2 Return to *return*

Finally, we reobtain *return* and the monadic structure of pipes in a slightly unusual way, by means of the *continuation* monad.

```
newtype Cont r a = Cont {runCont :: (a → r) → r}
instance Monad (Cont r) where
  return x = Cont (λk → k x)
  p >>= f = Cont (λk → runCont p (λx → runCont (f x) k))
```

If we specialize the result type r to $\text{Result}_R\ i\ o$, we get:

```
newtype ContP i o a = ContP ((a → ResultR i o) → ResultR i o)
```

The merge function for *ContP* is implemented in terms of $\text{merge}_{\text{Result}_R}$.

```
mergeCont :: ContP i m Void → ContP m o Void → ContP i o a
mergeCont (ContP p) (ContP q) = ContP (λk →
  mergeResultR (p absurd) (q absurd))
```

However, there is an issue: before $\text{merge}_{\text{Result}_R}$ can merge the two pipes, their continuations (the interpretations of the *return* constructor) must be supplied. Yet, the resulting pipe's continuation type k does not match that of either the upstream or downstream pipe. Thus we are stuck, unless we assume what we have been all along: that the two pipes are infinite. Indeed, in that case it does not matter that we don't have a continuation for them, as their continuation is never reached anyway. In short, $\text{merge}_{\text{Cont}}$ only works for never-returning pipes, which we signal with the return type *Void*, only inhabited by \perp .

4.3 Specialization for *IO*

To get exactly Spivey's representation, we instantiate the polymorphic type variable a in $\text{Result}_R\ i\ o$ to $\text{IO}\ ()$, which yields:

```
type Result i o = InCont i → OutCont o → IO ()
```

We can rewrite this type as a monad transformer stack, using two reader monad transformers for the two parameters.

```
newtype ReaderT r m a = ReaderT {runReaderT :: r → m a}
type Result' i o = ReaderT (InCont i) (ReaderT (OutCont o) IO) ()
```

Similarly, $\text{ContPipe}\ i\ o\ a$ can be written with a transformer stack by adding a *ContT* layer, since $\text{Cont}\ (m\ r)$ is equal to $\text{ContT}\ r\ m$ for any monad m .

```
newtype ContT r m a = ContT {runContT :: (a → m r) → m r}
type ContPipe' i o a =
  ContT () (ReaderT (InCont i) (ReaderT (OutCont o) IO)) a
```

This transformer stack view enables two additional useful operations: aborting the pipe and embedding an *IO* action. Both are specializations of generic functionality from the continuation monad transformer: *abort* and *lift_{ContT}*.

```

abort :: m r → ContT r m a
abort r = ContT (λk → r)

liftContT :: Monad m ⇒ m a → ContT r m a
liftContT p = ContT (λk → p >>= k)

exit' :: ContPipe' i o a
exit' = abort (liftReaderT (liftReaderT (return ())))

effect' :: IO a → ContPipe' i o a
effect' e = liftContT (liftReaderT (liftReaderT e))

```

5 Bidirectional Pipes

So far we have covered unidirectional pipes where information flows in one direction through the pipe, from the *output* operations in one pipe to the *input* operations in the next pipe downstream. However, some use cases also require information to flow upstream and pipes that support this are called bidirectional.

The *Proxy* data type at the core of the `pipes` library [5] implements bidirectional pipes. The operations *request* and *respond* are respectively downstream and upstream combinations of *input* and *output*. In addition, *Proxy* is also a monad transformer that embed effects of monad *m*.

```

data Proxy a' a b' b m r = Request a' (a → Proxy a' a b' b m r)
                          | Respond b (b' → Proxy a' a b' b m r)
                          | M          (m (Proxy a' a b' b m r))
                          | Pure r

```

We refer to the `pipes` source code [5] for the implementation of the corresponding *merge_{PL}* and *merge_{PR}* functions, which are called *+>>* and *>>~*.

We obtain a more efficient function-based representation by adapting the derivation of Sections 3 and 4. This yields the parameter type *PCPar*.

```

newtype PCPar i o a = PCPar (o → PCPar o i a → a)

```

The *Result_R* counterpart for *Proxy* takes two such *PCPar*s as input. In addition, the result type *r* is now a monadic type *m r* to be able to lift operations once it is wrapped with *Cont*.

```

type ProxyRep a' a b' b m = ∀ r. PCPar a a' (m r) →    -- request
                                PCPar b' b (m r) →    -- respond
                                m r

```

Then, we can proceed with defining the merge function for *ProxyRep* and the *Cont*-wrapped version similar to *Result_R*.

```

mergeProxyRep :: (c' → ProxyRep a' a c' c m) → ProxyRep c' c b' b m →
  ProxyRep a' a b' b m
mergeProxyRep fc' q = λ req res →
  let p' c' = fc' c' req
      q' = flip q res
  in q' (PCPar p')
newtype ContPr a' a b' b m r = ContPr { unContPr ::
  (r → ProxyRep a' a b' b m) → ProxyRep a' a b' b m }
mergeContPr :: (c' → ContPr a' a c' c m Void) →
  ContPr c' c b' b m Void → ContPr a' a b' b m r
mergeContPr fc' (ContPr q) = ContPr (λ k →
  mergeProxyRep (λ c' → unContPr (fc' c') absurd) (q absurd))

```

6 Benchmarks

Figure 1 shows the results of Spivey’s *primes* benchmark, which calculates the first n primes. The benchmarks are executed using the `criterion` library [9] on an Intel Core i7-6600U at 2.60 GHz with 8 GB memory running Ubuntu 16.04 and GHC 8.4.3, with `-O2` enabled.¹

The figure compares the `pipes` (v4.3.9) and `conduit` (v1.3.0.3) libraries to Spivey’s original implementation (`contpipe`) and our generalized form (`proxyrep`).

We can see that the former two libraries, which use an ADT representation, both show the quadratic performance behaviour for a use case with a high amount of merge steps. On the other hand, the latter two show the improved performance behaviour. The slight overhead of `proxyrep` compared to `contpipe` can be explained by the specialization to `IO ()` in the latter type.

Appendix B contains the results of some additional microbenchmarks.

7 Related Work

We have covered the main related works of Spivey [13], Shivers and Might [11] and the `pipes` library [5] in the body of the paper. Below we discuss some additional related work.

Encodings The Church [2,1] and Scott [8] encodings encode ADTs using functions. The encoding derived in this paper has a close connection to the Scott encoding. The Scott encoding for *Producer* and *Consumer* are *ScottP* and *ScottC*. By moving the quantified variable a to the definition, we obtain *SP* and *SC*.

```

newtype ScottP o = ScottP (∀ a. (o → ScottP o → a) → a)
newtype ScottC i = ScottC (∀ a. ((i → ScottC i) → a) → a)

```

¹ The benchmarks are at <https://github.com/rubenpieters/orth-pipes-bench>.




Fig. 1. Results of the *primes* benchmark.

$$\begin{aligned} \text{newtype } SP \ o \ a &= SP \ ((o \rightarrow SP \ o \ a \rightarrow a) \rightarrow a) \\ \text{newtype } SC \ i \ a &= SC \ (((i \rightarrow SC \ i \ a) \rightarrow a) \rightarrow a) \end{aligned}$$

Then, $\forall a. SP \ o \ a$ is representationally equivalent to *Producer_{Alt}* and similarly for $\forall a. SC \ i \ a$ and *Consumer_{Alt}* (see Appendix C).

If we look at the Scott encoding *ScottPipe_∞* for *Pipe_∞*, we can obtain an equivalent representation to *Result_R* by using *SP* and *SC* instead of *ScottPipe_∞* in the parameter corresponding to their operations.

$$\begin{aligned} \text{newtype } ScottPipe_{\infty} \ i \ o &= ScottPipe_{\infty} \\ &(\forall a. (o \rightarrow ScottPipe_{\infty} \ i \ o \rightarrow a) \rightarrow ((i \rightarrow ScottPipe_{\infty} \ i \ o) \rightarrow a) \rightarrow a) \\ \text{type } SP_{\infty} \ i \ o &= \forall a. (o \rightarrow SP \ o \ a \rightarrow a) \rightarrow ((i \rightarrow SC \ i \ a) \rightarrow a) \rightarrow a \end{aligned}$$

We dubbed this the *orthogonal encoding* due to the separation of the operations.

Conduit The *conduit* library [12] is another popular choice for Haskell stream processing. The two main differing points of *conduit* with *pipes* is a built-in representation of leftovers and detection of upstream finalization. Leftovers are operations representing unprocessed outputs. For example in a *takeWhile* pipe, which takes outputs until a condition is matched, the first element not matching the condition will also be consumed. This element can then be emitted as a leftover, which will be consumed by the downstream with priority. Detecting upstream finalization is handled by *input* returning *Maybe* values, where *Nothing* represents the finalization of the upstream.

Parsers Spivey mentioned in his work [13] that the *ContPipe* approach might be adapted to fit the use case of parallel parsers [3]. However, after gaining more insight into *ContPipe*, it does not seem that the merging operation for parsers immediately fits the pattern presented in this paper. One of the problematic elements is the *fail* operation, which is not passed as-is to the newly merged structure, but given a non-trivial interpretation. Namely, an interpretation dependent on the other structure during the recursive merge process.

Shallow To Deep Handlers The handlers framework by Kammar et al. [6] supports both shallow handlers, based on case analysis, and deep handlers, based on folds. They cover an example of transforming a producer and consumer merging function from shallow handlers to deep handlers. This example is related to our simplified setting in Section 3. To do this they introduce *Prod* and *Cons*, which are equivalent to our *ProdPar* and *ConsPar*. Compared to their example, we take a more step-by-step explanatory approach and additionally move to more complicated settings in our further sections.

Multihandlers The Frank language [7] is based on shallow handlers and supports a feature called multihandlers. These handlers operate on multiple inputs which have uninterpreted operations, much like pattern matching on multiple free structures. The patterns we have handled in this paper are concerned with pattern matching on multiple data structures and a mutual relation between these functions. This seems like an interesting connection to investigate further.

8 Conclusion

We have given an in-depth explanation of the principles behind the fast merging of the three-continuation approach. We have given a series of steps to derive this fast implementation from the less efficient one.

We apply this pattern to the setting of bidirectional pipes, as in the `pipes` library. This results in a more general version of this representation, but still has the same performance due to its efficient merge implementation.

We apply Spivey’s benchmarks [13] to check that our generalized encoding retains similar performance. We also include the `pipes` library in the benchmark to compare with a commonly used implementation of bidirectional pipes.

This pipes encoding has been made available as a library on github².

² <https://github.com/rubenpieters/Orthogonal-Pipes>

References

1. Böhm, C., Berarducci, A.: Automatic synthesis of typed lambda-programs on term algebras. *Theor. Comput. Sci.* **39**, 135–154 (1985). [https://doi.org/10.1016/0304-3975\(85\)90135-5](https://doi.org/10.1016/0304-3975(85)90135-5)
2. Church, A.: *The Calculi of Lambda-Conversion*. Princeton University Press, Princeton, New York (1941)
3. Claessen, K.: Parallel parsing processes. *J. Funct. Program.* **14**(6), 741–757 (Nov 2004). <https://doi.org/10.1017/S0956796804005192>
4. Ghani, N., Uustalu, T., Vene, V.: Build, augment and destroy, universally. In: Chin, W.N. (ed.) *Programming Languages and Systems*. pp. 327–347. Springer Berlin Heidelberg, Berlin, Heidelberg (2004)
5. Gonzalez, G.: Haskell *Pipes* library. <http://hackage.haskell.org/package/pipes> (2012)
6. Kammar, O., Lindley, S., Oury, N.: Handlers in action. In: *Proceedings of the 18th ACM SIGPLAN International Conference on Functional Programming*. pp. 145–158. ICFP '13, ACM, New York, NY, USA (2013). <https://doi.org/10.1145/2500365.2500590>
7. Lindley, S., McBride, C., McLaughlin, C.: Do be do be do. In: *Proceedings of the 44th ACM SIGPLAN Symposium on Principles of Programming Languages*. pp. 500–514. POPL 2017, ACM, New York, NY, USA (2017). <https://doi.org/10.1145/3009837.3009897>
8. Mogensen, T.: Efficient self-interpretation in lambda calculus. *Journal of Functional Programming* **2**(3), 345–364 (1992). <https://doi.org/10.1017/S0956796800000423>
9. O’Sullivan, B.: Haskell *Criterion* library. <http://hackage.haskell.org/package/criterion> (2009)
10. Plotkin, G.D., Abadi, M.: A logic for parametric polymorphism. In: *Proceedings of the International Conference on Typed Lambda Calculi and Applications*. pp. 361–375. TLCA '93, Springer-Verlag, London, UK, UK (1993), <http://dl.acm.org/citation.cfm?id=645891.671433>
11. Shivers, O., Might, M.: Continuations and transducer composition. In: *Proceedings of the 27th ACM SIGPLAN Conference on Programming Language Design and Implementation*. pp. 295–307. PLDI '06, ACM, New York, NY, USA (2006). <https://doi.org/10.1145/1133981.1134016>
12. Snoyman, M.: Haskell *Conduit* library. <http://hackage.haskell.org/package/conduit> (2011)
13. Spivey, M.: Faster coroutine pipelines. *Proc. ACM Program. Lang.* **1**(ICFP), 5:1–5:23 (Aug 2017). <https://doi.org/10.1145/3110249>

A Proof isomorphism *Producer* and *Producer_{Alt}*

We show the *Producer* case, an analogous strategy works for *Consumer*.

Theorem 1. *The types *Producer_{Alt}* *o* and *Producer* *o* are isomorphic.*

Proof. We show this by using $\mu x. G [x] = \forall x. (G [x] \rightarrow x) \rightarrow x$ [10].

$$\begin{aligned}
& \text{Producer}_{Alt} \ o \\
= & \quad (\text{see Appendix C}) \\
& \forall r. \ SP \ o \ r \\
= & \quad (\text{def. } SP) \\
& \forall r. \mu x. (o \rightarrow x \rightarrow r) \rightarrow r \\
= & \quad (\text{property } \mu) \\
& \forall r. \forall x. (((o \rightarrow x \rightarrow r) \rightarrow r) \rightarrow x) \rightarrow x \\
= & \quad (\text{swap } \forall) \\
& \forall x. \forall r. (((o \rightarrow x \rightarrow r) \rightarrow r) \rightarrow x) \rightarrow x \\
= & \quad (\text{property } \mu) \\
& \mu x. \forall r. (o \rightarrow x \rightarrow r) \rightarrow r \\
= & \quad (\text{uncurry}) \\
& \mu x. \forall r. ((o, x) \rightarrow r) \rightarrow r \\
= & \quad (\text{property } \mu) \\
& \mu x. \mu r. (o, x) \\
= & \quad (\text{drop unused } \mu) \\
& \mu x. (o, x) \\
= & \quad (\text{def. } Producer) \\
& Producer \ o \quad \square
\end{aligned}$$

An isomorphism consists of two inverses, here mr and mr^{-1} , for which $mr^{-1} \circ mr = id$ and $mr \circ mr^{-1} = id$ holds. Where mr is defined as:

$$\begin{aligned}
mr &:: Producer \ x \rightarrow Producer_{Alt} \ x \\
mr \ (Producer \ o \ r) \ (ConsPar \ h) &= h \ o \ (ProdPar \ (mr \ r))
\end{aligned}$$

And mr^{-1} is defined as:

$$\begin{aligned}
mr^{-1} &:: Producer_{Alt} \ x \rightarrow Producer \ x \\
mr^{-1} \ f &= f \ (ConsPar \ (\lambda x \ p \rightarrow Producer \ x \ (helpermr^{-1} \ p))) \\
helpermr^{-1} &:: ProdPar \ x \ (Producer \ x) \rightarrow Producer \ x \\
helpermr^{-1} \ (ProdPar \ f) &= f \ (ConsPar \ (\lambda x \ prod \rightarrow Producer \ x \ (helpermr^{-1} \ prod)))
\end{aligned}$$

We prove the first equation $mr^{-1} \circ mr = id$.

Theorem 2. $mr^{-1} \circ mr = id$.

Proof. We show this by equational reasoning.

$$\begin{aligned}
& mr^{-1} \circ mr \\
= & \text{(eta-expansion)} \\
& \lambda x \rightarrow mr^{-1} (mr\ x) \\
= & \text{(def. } mr) \\
& \lambda x \rightarrow mr^{-1} (\text{case } x \text{ of } (Producer\ o\ r) \rightarrow \lambda(ConsPar\ h) \rightarrow h\ o\ (ProdPar\ (mr\ r))) \\
= & \text{(case analysis on } x) \\
& \lambda(Producer\ o\ r) \rightarrow mr^{-1} (\lambda(ConsPar\ h) \rightarrow h\ o\ (ProdPar\ (mr\ r))) \\
= & \text{(def. } mr^{-1}) \\
& \lambda(Producer\ o\ r) \rightarrow (\lambda(ConsPar\ h) \rightarrow \\
& \quad h\ o\ (ProdPar\ (mr\ r))) (\lambda x\ p \rightarrow Producer\ x\ (helpermr^{-1}\ p))) \\
= & \text{(beta-reduction)} \\
& \lambda(Producer\ o\ r) \rightarrow Producer\ o\ (helpermr^{-1}\ (ProdPar\ (mr\ r))) \\
= & \text{(def. } helpermr^{-1}) \\
& \lambda(Producer\ o\ r) \rightarrow Producer\ o\ (mr\ r\ (ConsPar\ (\lambda x\ prod \rightarrow Producer\ x\ (helpermr^{-1}\ prod)))) \\
= & \text{(def. } mr^{-1}) \\
& \lambda(Producer\ o\ r) \rightarrow Producer\ o\ (mr^{-1}\ (mr\ r)) \\
= & \text{(coinduction hypothesis)} \\
& \lambda(Producer\ o\ r) \rightarrow Producer\ o\ r \\
= & \text{(def. } id) \\
& id
\end{aligned}$$

The second equation $mr \circ mr^{-1} = id$ is more complicated since we can not do case analysis on $Producer_{Alt}$, which is the function type $\forall a. ConsPar\ o\ a \rightarrow a$. We assume that values of this type can always be written using the constructor function $output_{Alt}$. Where $output_{Alt}$ is defined as:

$$\begin{aligned}
output_{Alt} &:: o \rightarrow Producer_{Alt}\ o \rightarrow Producer_{Alt}\ o \\
output_{Alt}\ o\ prod &= \lambda(ConsPar\ cons) \rightarrow cons\ o\ (ProdPar\ prod)
\end{aligned}$$

Conjecture 1. Values of $Producer_{Alt}$ can always be written in the form $output_{Alt}\ o\ r$.

This allows us to do a similar step to case analysis and complete the proof.

Theorem 3. $mr \circ mr^{-1} = id$.

Proof. We show this by equational reasoning.

$$\begin{aligned}
& mr \circ mr^{-1} \\
= & \text{(eta-expansion)} \\
& \lambda x \rightarrow mr\ (mr^{-1}\ x) \\
= & \text{(def. } mr^{-1}) \\
& \lambda x \rightarrow mr\ (x\ (ConsPar\ (\lambda x\ p \rightarrow Producer\ x\ (helpermr^{-1}\ p)))) \\
= & \text{(Conjecture 1)} \\
& \lambda(output_{Alt}\ o\ r) \rightarrow mr\ ((output_{Alt}\ o\ r)\ (ConsPar\ (\lambda x\ p \rightarrow Producer\ x\ (helpermr^{-1}\ p)))) \\
= & \text{(def. } output_{Alt}) \\
& \lambda(output_{Alt}\ o\ r) \rightarrow mr\ ((\lambda(ConsPar\ cons) \rightarrow
\end{aligned}$$

$$\begin{aligned}
& \text{cons } o \text{ } (\text{ProdPar } r)) \text{ } (\text{ConsPar } (\lambda x \text{ } p \rightarrow \text{Producer } x \text{ } (\text{helpermr}^{-1} \text{ } p)))) \\
= & \text{ (beta-reduction)} \\
& \lambda(\text{output}_{\text{Alt}} \text{ } o \text{ } r) \rightarrow \text{mr } ((\lambda x \text{ } p \rightarrow \text{Producer } x \text{ } (\text{helpermr}^{-1} \text{ } p)) \text{ } o \text{ } (\text{ProdPar } r)) \\
= & \text{ (beta-reduction)} \\
& \lambda(\text{output}_{\text{Alt}} \text{ } o \text{ } r) \rightarrow \text{mr } (\text{Producer } o \text{ } (\text{helpermr}^{-1} \text{ } (\text{ProdPar } r))) \\
= & \text{ (def. } \text{helpermr}^{-1} \text{)} \\
& \lambda(\text{output}_{\text{Alt}} \text{ } o \text{ } r) \rightarrow \text{mr } (\text{Producer } o \text{ } (r \text{ } (\text{ConsPar } (\lambda x \text{ } \text{prod} \rightarrow \text{Producer } x \text{ } (\text{helpermr}^{-1} \text{ } \text{prod})))))) \\
= & \text{ (def. } \text{mr}^{-1} \text{)} \\
& \lambda(\text{output}_{\text{Alt}} \text{ } o \text{ } r) \rightarrow \text{mr } (\text{Producer } o \text{ } (\text{mr}^{-1} \text{ } r)) \\
= & \text{ (def. } \text{mr} \text{)} \\
& \lambda(\text{output}_{\text{Alt}} \text{ } o \text{ } r) \rightarrow \lambda(\text{ConsPar } h) \rightarrow h \text{ } o \text{ } (\text{ProdPar } (\text{mr } (\text{mr}^{-1} \text{ } r))) \\
= & \text{ (coinduction hypothesis)} \\
& \lambda(\text{output}_{\text{Alt}} \text{ } o \text{ } r) \rightarrow \lambda(\text{ConsPar } h) \rightarrow h \text{ } o \text{ } (\text{ProdPar } r) \\
= & \text{ (def. } \text{output}_{\text{Alt}} \text{)} \\
& \lambda(\text{output}_{\text{Alt}} \text{ } o \text{ } r) \rightarrow \text{output}_{\text{Alt}} \text{ } o \text{ } r \\
= & \text{ (def. } \text{id} \text{)} \\
& \text{id}
\end{aligned}$$

B Additional Benchmarks

This section contains some additional benchmark results.

B.1 Deep-Pipe

The results of Spivey’s deep-pipe benchmark. This is a microbenchmark of the merge operator, essentially a more extreme version of the *primes* benchmark. The x-axis is logarithmic for readability of the graph.




Fig. 2. Results of the deep-pipe n benchmark.

B.2 Deep-Seq

The results of Spivey’s deep-seq benchmark. This is a microbenchmark of the bind operator. The **conduit** library exhibits better performance compared to **pipes** in this benchmark since it uses a **Codensity** encoding to improve the performance of bind. The x-axis is logarithmic for readability of the graph.




Fig. 3. Results of the deep-seq n benchmark.

B.3 streaming-benchmarks

The results of some selected operation microbenchmarks from **streaming-benchmarks** (<https://github.com/composewell/streaming-benchmarks/>). These benchmarks measure the performance of specific operations such as *map* and *fold*, the continuation-based implementation performs worse compared to the **pipes** and **conduit** libraries. Despite this, the continuation-based approach is much more performant in the use case of many merges as seen in the previous benchmarks.




Fig. 4. Results of various microbenchmarks.

C Producer_{Alt} is representationally equal to $\forall a.SP\ o\ a$

We show the Producer case, an analogous strategy works for Consumer .

Theorem 4. *The representation Producer_{Alt} is \sim_R (representationally equivalent) to $\forall a.SP\ o\ a$.*

The relevant definitions for the proof are:

newtype $\text{ConsPar}\ x\ a = \text{ConsPar}\ (x \rightarrow \text{ProdPar}\ x\ a \rightarrow a)$
newtype $\text{ProdPar}\ x\ a = \text{ProdPar}\ (\text{ConsPar}\ x\ a \rightarrow a)$
newtype $SP\ o\ a = SP\ ((o \rightarrow SP\ o\ a \rightarrow a) \rightarrow a)$
type $\text{Producer}_{Alt}\ o = \forall a.\text{ConsPar}\ o\ a \rightarrow a$

Proof. First, we show that $\text{ConsPar}\ o\ a \rightarrow a \sim_R SP\ o\ a$.

$$\begin{aligned} & \text{ConsPar}\ o\ a \rightarrow a \\ \sim_R & \text{ (def. } \text{ConsPar}) \\ & (o \rightarrow \text{ProdPar}\ o\ a \rightarrow a) \rightarrow a \\ \sim_R & \text{ (def. } \text{ProdPar}) \\ & (o \rightarrow (\text{ConsPar}\ o\ a \rightarrow a) \rightarrow a) \rightarrow a \\ \sim_R & \text{ (coinduction hypothesis)} \\ & (o \rightarrow SP\ o\ a \rightarrow a) \rightarrow a \\ \sim_R & \text{ (def. } SP) \\ & SP\ o\ a \end{aligned}$$

Then,

$$\begin{aligned} & \text{Producer}_{Alt}\ o \\ \sim_R & \text{ (def. } \text{Producer}_{Alt}) \\ & \forall a.\text{ConsPar}\ o\ a \rightarrow a \\ \sim_R & \text{ (proven above)} \\ & \forall a.SP\ o\ a \quad \square \end{aligned}$$